Tensorial analysis of second-order elastic constants

Input: the stiffness matrix in Voigt notation

The matrix below is that obtained for MIL-47(V).

In[1]:=
```
CVoigt = (40.693  12.578  9.276   0       0      0
          12.578  62.597  46.976  0       0      0
          9.276   46.976  36.149  0       0      0
          0       0       0       50.827  0      0
          0       0       0       0       7.756  0
          0       0       0       0       0      9.304);
```

```
Eigenvalues[CVoigt] // Reverse
```

Out[2]=
```
{0.570112, 7.756, 9.304, 36.7203, 50.827, 102.149}
```

The eigenvalues above should all be positive, indicating stability of the crystal.

Code for tensorial analysis & calculation of mechanical properties

The block below contains all the code of the notebook. We use it in the next section to plot properties of the material.

```
In[3]:= SVoigt = Inverse[CVoigt];
VoigtMat = {{1, 6, 5}, {6, 2, 4}, {5, 4, 3}};
SVoigtCoeff[p_, q_] := 1 / (Ceiling[p / 3] * Ceiling[q / 3]);
Smat = Table[SVoigtCoeff[VoigtMat[[i, j]], VoigtMat[[k, l]]]
    * SVoigt[[VoigtMat[[i, j]], VoigtMat[[k, l]]]],
   {i, 1, 3}, {j, 1, 3}, {k, 1, 3}, {l, 1, 3}];
Cmat = Table[CVoigt[[VoigtMat[[i, j]], VoigtMat[[k, l]]]],
   {i, 1, 3}, {j, 1, 3}, {k, 1, 3}, {l, 1, 3}];
dirVector[θ_, ϕ_] := {Sin[θ] * Cos[ϕ], Sin[θ] * Sin[ϕ], Cos[θ]};
YoungModulus[θ_, ϕ_] := Module[{a},
   a = dirVector[θ, ϕ];
   1 / Sum[a[[i]] * a[[j]] * a[[k]] * a[[l]] * Smat[[i, j, k, l]],
     {i, 1, 3}, {j, 1, 3}, {k, 1, 3}, {l, 1, 3}]
  ];
linearCompressibility[θ_, ϕ_] := Module[{a},
   a = dirVector[θ, ϕ];
   Sum[a[[i]] * a[[j]] * Smat[[i, j, k, k]],
    {i, 1, 3}, {j, 1, 3}, {k, 1, 3}]
  ];
dirVector2[θ_, ϕ_, χ_] := {Cos[θ] * Cos[ϕ] * Cos[χ] - Sin[ϕ] * Sin[χ],
   Cos[θ] * Sin[ϕ] * Cos[χ] + Cos[ϕ] * Sin[χ], -Sin[θ] * Cos[χ]};
shearModulus[θ_, ϕ_, χ_] := Module[{a, b},
   a = dirVector[θ, ϕ];
   b = dirVector2[θ, ϕ, χ];
   1 / (4 * Sum[a[[i]] * b[[j]] * a[[k]] * b[[l]] * Smat[[i, j, k, l]],
       {i, 1, 3}, {j, 1, 3}, {k, 1, 3}, {l, 1, 3}])
  ];
PoissonRatio[θ_, ϕ_, χ_] := Module[{a, b},
   a = dirVector[θ, ϕ];
   b = dirVector2[θ, ϕ, χ];
   -Sum[a[[i]] * a[[j]] * b[[k]] * b[[l]] * Smat[[i, j, k, l]],
      {i, 1, 3}, {j, 1, 3}, {k, 1, 3}, {l, 1, 3}]
    / Sum[a[[i]] * a[[j]] * a[[k]] * a[[l]] * Smat[[i, j, k, l]],
     {i, 1, 3}, {j, 1, 3}, {k, 1, 3}, {l, 1, 3}]
  ];
```

Examples of use of the analysis code

First, we focus on directional Young’s modulus. We calculate its minimal and maximal values.

```
In[14]:= Maximize[YoungModulus[θ, ϕ], {θ, ϕ}]
Out[14]= {96.593, {θ → 2.28741, ϕ → 1.5708}}
```

```
In[15]:= Minimize[YoungModulus[θ, ϕ], {θ, ϕ}]
```

Out[15]= $\{0.895102, \{\theta \to -8.27181\times 10^{-25}, \phi \to -0.447578\}\}$

Then we do a 3D plot:

In[16]:=
```
SphericalPlot3D[YoungModulus[θ, ϕ], θ, ϕ,
 PlotRange → All, AxesLabel → {"x", "y", "z"}, PlotPoints → 40, Mesh → 30]
```

Out[16]=



And finally, a polar plot in the plane of flexibility (*zy*):

In[17]:=
```
PolarPlot[YoungModulus[θ, 0], {θ, 0, 2 π}, PlotRange → All, PlotStyle → Thick]
```

Out[17]=



Now, we turn to the linear compressibility. We start with a 3D plot (green lobe for positive LC, red lobe for negative LC):

In[18]:=
```
Show[
 SphericalPlot3D[Max[linearCompressibility[θ, ϕ], 0], θ, ϕ,
  PlotStyle → Directive[Green, Opacity[0.7]], PlotRange → All],
 SphericalPlot3D[Max[-linearCompressibility[θ, ϕ], 0],
  θ, ϕ, PlotStyle → Directive[Red, Opacity[0.7]]],
 PlotRange → All, AxesLabel → {x, y, z}]
```

Out[18]=



And we characterize the value along each crystallographic axis:

In[19]:=
```
{linearCompressibility[π / 2, 0],
 linearCompressibility[π / 2, π / 2], linearCompressibility[0, 0]}
```

Out[19]= {0.022145, -0.20056, 0.28261}

Finally, we characterize the minimal and maximal values (and corresponding directions) for shear modulus and Poisson's ratio:

In[20]:=
```
Minimize[shearModulus[θ, ϕ, χ], {θ, ϕ, χ}]
```

Out[20]= $\{0.290397, \{\theta \to 0.785398, \phi \to 1.5708, \chi \to -2.86758 \times 10^{-20}\}\}$

In[21]:=
```
Maximize[shearModulus[θ, ϕ, χ], {θ, ϕ, χ}]
```

Out[21]= $\{50.827, \{\theta \to 1.29247 \times 10^{-26}, \phi \to 0.00931826, \chi \to -1.58011\}\}$

In[22]:=
```
Minimize[PoissonRatio[θ, ϕ, χ], {θ, ϕ, χ}]
```

Out[22]= $\{-2.1777, \{\theta \to 1.2049, \phi \to -6.77937 \times 10^{-23}, \chi \to 8.29536 \times 10^{-24}\}\}$

```
In[23]:= Maximize[PoissonRatio[θ, ϕ, χ], {θ, ϕ, χ}]
```

Out[23]= $\{2.16127, \{\theta \to -1.17537, \phi \to 5.908 \times 10^{-25}, \chi \to -1.5708\}\}$